Programowanie w Logice

Przeszukiwanie rozwiązań

Przemysław Kobylański

Przeszukiwanie rozwiązań

Generowanie wszystkich rozwiązań

- Prolog nie tylko potrafi **sprawdzić czy** dana spełnia warunek ale również potrafi **wygenerować wszystkie** dane spełniające warunek.
- Jest to możliwe dzięki temu, że Prolog wyraża **relacje** a nie tylko **funkcje** (dopuszcza dowolne przepływy danych).

Przeszukiwanie rozwiązań

Generowanie wszystkich rozwiązań

```
% perm(L1, L2) zachodzi, gdy
% lista L2 jest permutacja listy L1.

perm([], []).
perm(L1, [X | L3]) :-
        select(X, L1, L2),
        perm(L2, L3).
```

Przeszukiwanie rozwiązań

Generowanie wszystkich rozwiązań

Generowanie listy `[X | L3]` będącej permutacją danej listy `L1`:



Wybór elementu `X` na tyle sposobów jaka jest długość listy `L1`.

Przeszukiwanie rozwiązań

Generowanie wszystkich rozwiązań

```
?- perm([1, 2, 3], [3, 1, 2]).
true .

?- perm([1, 2, 3], [3, 2, 2]).
false.

?- perm([1, 2, 3], X).
X = [1, 2, 3] ;
X = [1, 3, 2] ;
X = [2, 1, 3] ;
X = [2, 3, 1] ;
X = [3, 1, 2] ;
X = [3, 2, 1] ;
false.
```

Przeszukiwanie rozwiązań

Generowanie i testowanie

- Niech `rozwiązanie(X)` będzie warunkiem generującym wszystkie rozwiązania.
- Niech `dobre(X)` będzie warunkiem sprawdzającym czy rozwiązanie jest dobre.
- Wówczas warunek `rozwiązanie(X), dobre(X)` generuje wszystkie dobre rozwiązania.
- Generowanie odbywa się zgodnie ze schematem następującej pętli:



Przeszukiwanie rozwiązań

Generowanie i testowanie

- Zaprezentujemy program znajdujący ustawienie `N` hetmanów na szachownicy o `N` wierszach i `N` kolumnach.
- Ustawienie będziemy kodować w postaci permutacji listy liczb od `1` do `N`.
- Na *i*-tej pozycji takiej permutacji zapisany będzie numer wiersza, w którym stoi hetman z *i*-tej kolumny.
- Gdy ustawimy hetmany zgodnie z permutacją, to żadne dwa nie będą się biły w kolumnie i w wierszu.
- Pozostaje do sprawdzenia, czy żadne dwa nie biją się po ukosie.

Przeszukiwanie rozwiązań

Generowanie i testowanie

```
dobra(X) :-
        \+ zla(X).

% zla(X) zachodzi, gdy wsrod hetmanow ustawiony
% zgodnie z permutacja X sa dwa ktore sie bija

zla(X) :-
        append(_, [Wi | L1], X),
        append(L2, [Wj | _], L1),
        length(L2, K),
        abs(Wi - Wj) =:= K + 1.

%    abs(Wi - Wj) = odleglosc w pionie
%          K + 1 = odleglosc w poziomie
```

Przeszukiwanie rozwiązań

Generowanie i testowanie

```
% hetmany(N, P) zachodzi, gdy permutacja P
% koduje poprawne ustawienie N hetmanow

hetmany(N, P) :-
        numlist(1, N, L),
        perm(L, P),
        dobra(P).
```

Przeszukiwanie rozwiązań

Generowanie i testowanie

```
?- hetmany(4, X).
X = [2, 4, 1, 3] ;
X = [3, 1, 4, 2] ;
false.

?- findall(X, hetmany(8, X), L), length(L, N).
L = [[1, 5, 8, 6, 3, 7, 2, 4], [...|...]|...],
N = 92.
```

Predykat `findall(X, p(X), L)` tworzy listę `L` wszystkich wartości `X` spełniających warunek `p(X)`.

Przeszukiwanie rozwiązań

Odcięcie

- Jeśli warunek dostarcza wiele wartości, to stawiając po nim wykrzyknik (odcięcie), zostanie znaleziona tylko pierwsza z nich (odcinamy poszukiwanie kolejnych).
- Predykat ! jest zawsze spełniony ale wpływa na poszukiwanie rozwiązać i nie dopuszcza do wykonania nawrotu celem szukania kolejnego rozwiązania.

```
?- member(X, [1, 2, 3]), X > 1.5.
X = 2 ;
X = 3.

?- member(X, [1, 2, 3]), !, X > 1.5.
false.
```

Przeszukiwanie rozwiązań

Odcięcie



Odcięcie działa w celu jak „błona półprzepuszczalna”: pozwala przejść z lewa na prawo ale nie odwrotnie.

Przeszukiwanie rozwiązań

Odcięcie

Zupełnie inaczej działa odcięcie jeśli użyto go w ciele reguły.

Example (Algorytm Świętego Mikołaja <wersja 1>)

Święty Mikołaj stosuje poniższe reguły do rozstrzygnięcia czy dziecko `X` dostanie nagrodę `Y` albo karę `Y`.

```
dostanie(X, Y) :- grzeczne(X), nagroda(Y).
dostanie(X, Y) :- \+ grzeczne(X), kara(Y).
```

- W powyższym przykładzie, jeśli dziecko `X` nie było grzeczne, to warunek `grzeczne(X)` jest sprawdzany dwukrotnie.
- W takim przypadku zawodzi warunek `grzeczne(X)` z pierwszej reguły predykatu `dostanie(X, Y)` i Prolog sięga do drugiej reguły.
- W drugiej regule sprawdzane jest ponownie czy dziecko było grzeczne i jeśli warunek ten zawodzi, to wybierana jest dla niego kara `Y`.

Przeszukiwanie rozwiązań

Odcięcie

Example (Algorytm Świętego Mikołaja <wersja 2>)

```
dostanie(X, Y) :- grzeczne(X), nagroda(Y).
dostanie(X, Y) :- kara(Y).
```

- Ta wersja nie jest poprawna.
- Jeśli będziemy próbowali wywnioskować wszystko co może dostać grzeczne dziecko, to po wyczerpaniu wszystkich nagród, predykat stwierdzi na podstawie drugiej reguły, że należy dawać mu kolejne kary.

Przeszukiwanie rozwiązań

Odcięcie

Example (Algorytm Świętego Mikołaja <wersja 3>)

```
dostanie(X, Y) :- grzeczne(X), !, nagroda(Y).
dostanie(X, Y) :- kara(Y).
```

- Ta wersja jest poprawna.
- Jeśli dziecko `X` jest grzeczne, to pierwsza reguła odpowie jakie nagrody może ono dostać.
- Jeśli wyczerpią się już wszystkie nagrody, to przy próbie wycofania się do sprawdzania czy dziecko jest grzeczne, odcięcie ! spowoduje niepowodzenie i porzucenie dalszego sprawdzania warunku `dostanie(X, Y)`, nawet jeśli są jeszcze jakieś inne reguły w jego definicji.

Przeszukiwanie rozwiązań

Przykłady zastosowań odcięcia

Example (`once/1`)

Predykat `once(Goal)` znajduje pierwszą odpowiedź na cel `Goal` i nie będzie szukał kolejnych.

```
once(Goal) :-
        Goal, !.
```

Przeszukiwanie rozwiązań

Przykłady zastosowań odcięcia

Example (Negacja)

Predykat `\+ Goal` zawodzi gdy cel `Goal` jest spełniony. W przeciwnym przypadku jest spełniony.

```
\+ Goal :-
        Goal,
        !,
        fail.
\+ _.
```

Przeszukiwanie rozwiązań

Przykłady zastosowań odcięcia

Example (`forall/2`)

Predykat `forall(Generator, Test)` sprawdza czy wszystkie dane generowane warunkiem `Generator` spełniają również warunek `Test`.

```
forall(Generator, Test) :-
    \+ (Generator, \+ Test).
```

Przeszukiwanie rozwiązań

Przykłady zastosowań odcięcia

Example (Własna wersja `var/1`)

```
myvar(X) :-
        \+ \+ X = a,
        \+ \+ X = b.
```

- Pierwszy warunek `\+ \+ X = a` jest spełniony tylko gdy `X` jest zmienną (unifikuje się z czymkolwiek) lub jest stałą `a`, przy czym jeśli `X` jest zmienną to nic nie zostanie pod nią podstawione.
- Drugi warunek `\+ \+ X = a` jest spełniony tylko gdy `X` jest zmienną (unifikuje się z czymkolwiek) lub jest stałą `b`, przy czym jeśli `X` jest zmienną to nic nie zostanie pod nią podstawione.
- Zatem oba warunki spełnione są tylko gdy `X` jest zmienną i nic nie zostanie pod nią podstawione.